

Shortest Path example in SAS

This example defines a weighted graph inside the SAS code. The first shortestpath call finds the path of minimum weight from vertex *A* to vertex *D*. The output is written to the file ShortPath_AtoD in the work library. The second shortestpath call finds the path of minimum weight from every possible source to every possible sink. These paths are written to the file ShortPathP and the weight of each path is written to the file ShortPathW. Finally, the third shortestpath call computes the shortest path from source *A* to all possible sinks. Output is written to the file All_ShortPath_SourceA.

```
* shortest path examples;
* input a weighted graph in SAS code;
data LinkSetIn;
input from $ to $ weight @@;
datalines;
A B 3
A C 2
A D 6
A E 4
B D 5
B F 5
C E 1
D E 2
D F 1
E F 4
;
* find the shortest path from A (source) to D (sink);
proc optgraph data_links = LinkSetIn;
shortpath
source = A
sink = D
out_paths = ShortPath_AtoD;
* write output of the shortest path from A to D to the file
ShortPath_AtoD in your work library;
run;
* find all shortest paths using all source to sink combinations;
proc optgraph data_links = LinkSetIn;
shortpath
out_weights = ShortPathW
out_paths = ShortPathP;
* write the total weight of the shortest path to ShortPathW in your
work library
write the actual paths to the file ShortPathP in your work library;
run;
*perhaps vertex A represents a warehouse and you want all shortest
paths from source A to all possible sinks;
data NodeSubSetIn;
input node $ source;
datalines;
```

```
A 1
;
proc optgraph
data_nodes_sub = NodeSubSetIn
data_links = LinkSetIn;
shortpath out_paths = All_ShortPath_SourceA;
* pipe all shortest paths from A to file All_ShortPath_SourceA in work
library;
run;
```